intusoft

(213) 833-0710

Personal Computer Circuit Design Tools

N E W S L E T T E R

Introducing IS_SPICE/386 *Faster than a VAX® 11/780*

Hidden within 80386 based computers is a full 32 bit architecture that breaks 640 Kbyte program size constraints and can speed program execution by a factor of 4. However, the 80386 must be run in protected mode to unlock this latent speed. Recent extensions to the Microsoft/IBM® DOS allows full access to DOS functions by a protected mode program. IS_SPICE/386 uses these new techniques to bring mainframe power to your 80386 based PC. Turn to page 10-10 for speed comparisons.

In This Issue

Updates for the Summer

IS_SPICE version 1.41 will replace vers. 1.0 beginning July 1. This is a real mode version of SPICE compatible with all IBM equivalent PC's. The RAM requirement is reduced by 25K Bytes and numerical exception processing is improved so that real indefinite errors no longer occur during distortion analysis. Data storage is dependent on the number of output points rather than the number of iterations, making long transient runs more feasible. The x axis variable has 6 digit precision and a .PRINT TRAN control is no longer mandatory when .PLOT TRAN is invoked.

Simulations with SPICE, our new IS_SPICE manual will be available in August. The book combines our previous newsletters with the IS_SPICE users manual and material from our SPICE class. You will receive a copy free by returning the registration card shipped with your update or with the purchase of either version 1.41 or IS_SPICE/386.

VAX is the trademark, ®, of Digital Equipment Corp., IBM is the trademark, ®, of International Business Machines, Inc.

Modeling Thermal Effects

Temperature in SPICE is handled as a built-in constant. Any circuit in which temperature interacts dynamically requires a different approach. Circuits in this class would include those using thermistors, bipolar power output stages and controllers that use temperature as a state variable.

SPICE elements can be connected to make an electrical analog of a heat transfer process. The analog used here makes the following equivalences:

Electrical Parameter	**Thermal Parameter**
Voltage	Temperature in Degrees Kelvin
Current	Power in Watts
Resistance	Thermal Resistance (Deg C or K per Watt
Capacitance	Thermal capacity in Watt-Sec per Degree C or K

Scaling laws could have been applied to align the range of electrical and thermal parameters, perhaps improving computational efficiency. For example, current in milliamps could represent watts. We have chosen not to perform this scaling in order to simplify the user interface.

The Thermal Node

For each electrical element, there will be a thermal node with the following internal equivalent circuit.



Figure 1; TCELL, A Thermal Node Equivalent Circuit

Notice that all modes of heat transfer (conduction, convection and radiation) are lumped together. A single thermal resistance and boundary temperature are used. Thermal nodes can be connected using thermal resistance and power can be added with current generators. Bringing the thermal node through to the main circuit allows simulations needing initial conditions to use the .IC statement and UIC keyword.

Modeling a Thermistor

A thermistor model was selected to illustrate this principal. Thermistors find a variety of applications in electronics. Their high initial resistance can be used for transient surge suppression. The exponential sensitivity makes them ideal for control system sensors, including fluid flow as well as the more obvious temperature control. Self heating is critical in predicting circuit operation. The following equation describes thermistor resistance as a function of temperature.

$$R = R0e^{\beta(1/T - 1/T0)} \qquad \text{Equation 1.0}$$

Figure 2 illustrates the thermal and electrical model with elements from the PRE_SPICE system library. The new EXP model is shown on the next page. The thermistor model was tested using a voltage step and series resistor. Current was measured as a function of time. After debugging the model, it was turned into a subcircuit with the following parameter list:

R0	Resistance at 300 Deg. Kelvin.
BETA	Coefficient, β, from equation 1.
HC	Thermal resistance from the thermal node to the boundary, TINF.
TAU	Thermal time constant.
TINF	Thermal boundary temperature in Deg. Kelvin.

The thermal time constant applies only to the local thermal mass, it will change as other components are connected.

Modeling an Exponential Function

An exponential can be modeled using a voltage controlled voltage source with a polynomial expansion or by adapting the diode model within SPICE. Polynomial expansions will fail for large inputs. Moreover, the modern computer algorithm for exponentials is considerable faster. The SPICE diode equation is:

$$i = Is(e^{V/NVt} - 1)$$

The following equivalent circuit makes an output voltage proportional to exp(input). In subtracting the offset at zero voltage, a small numerical error is introduced that limits the usefulness for negative voltages. The following SPICE circuit listing along with selected tabular output illustrates the new EXP function.



Listing

```
.SUBCKT EXP 1 2
R  1 0 1MEG
E 3 0 1 0 .0258642
V 3 4
RD 4 0 1MEG
D 4 0 DIODE
.MODEL DIODE D(IS=1)
I 0 4 -1
H 2 0 V 1
.ENDS
```

Test Results

Input	Output	Error in PPM
0.00000E+00	1.000000E+00	0.0
4.00000E-01	1.491825E+00	0.2
3.60000E+00	3.659830E+01	1.8
6.80000E+00	8.978505E+02	3.6
1.00000E+01	2.202658E+04	5.2
1.32000E+01	5.403687E+05	7.0
1.64000E+01	1.325663E+07	8.4
1.96000E+01	3.252193E+08	10.2

Figure 2, Exponential Function Test Results



Figure 3, A System Level Thermistor Model

Next, Figure 4 shows the new thermistor symbol and its listing. Initializing the dividers takes more iterations than normal, so it is necessary to set ITL1=500 in the .OPTIONS control statement.

```
.SUBCKT THRMR 10  3  9
G1  6  3  7  0  1
V1  5  0  {BETA}
X1  5  9  4  DIVIDE
X2  4  2  EXP
X3  1  2  7 DIVIDE
*TCELL
VTC1  21  0 {TINF}
CTC1  9  0  {TAU/HC}
RTC1  9  21  {HC}
GX5  0  9  POLY(2)  10  0  7  0  0  0  0  0  1.0000
V2  10  6
E1  1  0  3  6  {-(2.71828^(BETA/300)/R0)}
.ENDS
.SUBCKT DIVIDE  1  2  4
* V4 = V1 / V2
R1  1  0  1MEG
R2  2  0  1MEG
R4  4  0  1MEG
G1  0  3  1  0  1
G2  3  0  POLY(2)  2  0  3  0  0  0  0  0  1
R3  3  0  100MEG
E1  4  0  3  0  1
.ENDS
.SUBCKT EXP  1  2
R  1  0  1K
E  3  0  1  0   .0258642
V  3  4
RD  4  0  1MEG
D  4  0  DIODE
.MODEL  DIODE  D(IS=1)
I  0  4  -1
H  2  0  V  1
.ENDS
```




```
TRMTST, Thermister test circuit
*SPICE_NET
.TRAN  .05  10
*INCLUDE THERMAL.LIB
.OPTIONS ACCT ITL1=500
.PRINT TRAN  V(2)  I(V1) V(3)
R1  1  2  1.5K
X5  2  0  3 THRMR  {R0=40K BETA=4K
+  TAU=1.5 HC=500 TINF=300 }
V1  1  0  PULSE  0  50
.END
```

Figure 4, A Thermistor Model with Heat Transfer

Dynamic Thermal Models for Junction Transistors

The operating point of bipolar junction transistors, BJT’s, frequently depends on thermal properties of the electrical package. There are two main temperature dependencies, the base-emitter voltage and collector-base leakage current. While these are modeled accurately within SPICE, they are taken as constant values for each simulation. The dynamic thermal modeling technique can be applied to BJT’s and diodes using the techniques described here. Figure 5 shows how the temperature dependencies can be introduced.




```
.SUBCKT TNBJT 7 16 10 4
V1 7 1
V2 6 2
H1 3 0 V1 1
RH1 3 0 1MEG
H2 5 0 V2 1
RH2 5 0 1MEG
EX2 13 0 POLY(2) 7 10 3 0 0 0 0 0 1
RX2 13 0 1MEG
EX3 8 0 POLY(2) 5 0 16 10 0 0 0 0 1
RX3 8 0 1MEG
EX4 9 0 POLY(2) 13 0 8 0 0 1 1
RX4 9 0 1MEG
G1 0 4 9 0 1
R1 4 11 {HC}
C1 4 0 {TAU/HC}
V3 11 0 {TINF}
*ICBO SET TO DOUBLE EVERY 10 Deg. C
*Vbe decreases 2mv per Deg. C
EEXP 40 0 4 0 {.0258642*6.93E-2}
RX5 40 0 1MEG
X5 40 12 EXPP
G2 7 16 12 0 {ICBO*9.35}
E1 16 6 0 4 .6 2M
Q1 1 2 10 NBJT
.MODEL NBJT NPN(CJC={2.2*COB} TF={.16/FT} TR={1.7*TS}
+ CJE={7*COB} RC={.5/IMAX} VAF=100 IKF={.7*IMAX} IS={2E-15*IMAX} )
.ENDS
.SUBCKT EXPP 3 2
R 2 0 1MEG
V 3 4
RD 4 0 1MEG
D 4 0 DIODE
.MODEL DIODE D(IS=1E-10)
I 0 4 -1E-10
H 2 0 V 1
.ENDS
```

Figure 5, Thermal model for an NPN Bipolar Junction Transistor

This model will be applied to a complementary class AB power output stage in our next newsletter.

IS_SPICE/386 Performance Comparisons

The benchmark circuits shown here were used to compare hardware and software performance. The hardware configuration is described using shorthand notation for 386 machines, the number following the slash identifies the coprocessor type. The speed is given in MegHz after the dash. Software with the IS prefix refers to the Intusoft version number. The elliptic filter was taken from our August 1986 newsletter. The Div4 circuit is a 2 stage digital frequency divider. The correlator was heretofore a mainframe application; requiring over 1 MegaByte just to hold the equation matrix and temporary output. With these benchmark circuits you can look at your current hardware and software configuration and extrapolate performance changes.

Hardware		PC-AT	
Software		IS1.41	IS1.41
Circuit	**Memory** (bytes)		
Elliptic Filter	68,016	537	262
Div4	84,920	1,771	956
Correlator	1,025,808	too big	too big
Speed Index		.20	.39

Note: The correlator was used to compute the speed index where possible. The speed index for a VAX8820 was measured as 6.85.

Memory usage has been converted from 32 bit words, reported by SPICE, to bytes; it is the storage required for solving the matrix of equations, holding model parameters and temporary outputs. IS_SPICE version 1.xx reserves 160K Bytes while the VAX and /386 versions reserve 1.6MegaBytes.

Why not use OS/2 to break the 640K RAM Barrier

The IBM PC's disk operating system and ROM BIOS use interrupts that Intel had reserved for future machines. When the 80286 was introduced, the address space available in protected mode could not be used. Three years later, OS/2 solved the problem, but limited the solution to the 80286 architecture. While OS/2 runs on 386 machines, it handicaps the full potential up to a factor of 4 in speed. Programs that previously ran under DOS must be recompiled for OS/2. Most of these programs run 15% to 20% slower because of the virtual memory addressing scheme and the concurrent operating system overhead. Large SPICE circuits will run even slower.

Even more astounding than the circuit size gains we have all awaited, is the simulation speed improvement.

Standard Benchmark Circuits

	386/287-16		386/387-16		VAX11/780	386/1167-20
	IS1.41	IS/386	IS1.41	IS/386	UCB2G.6	IS/386
	Execution Time in Seconds					
	262	136	193	68	----	43
	956	615	519	216	----	132
	too big	26,844	too big	9,048	14,611	5,475
	.39	.55	.62	1.61	1.0	2.7

The 80287 and 80387 coprocessors are made by Intel. The 1167 is made by Weitek, Inc.

Benchmark Circuits Summary Data

	Nodes	Elements	Transistors	AC	DC	TRAN
Elliptic Filter	54	118	15	x	x	x
Div4	107	256	30			x
Correlator	1162	2967	305			x

Why Large SPICE Circuits Run Slow with 80286 Based Protected Mode Software

SPICE solves equations by matrix reduction. In this process, matrix pointers are manipulated rather than the actual matrix data. The access of data within the memory will not follow the usual premise that successive memory addresses are nearby. The 80286 architecture is still bound by 64K Byte physical segments. When a new segment is selected, the physical address associated with that segment must be fetched from a memory descriptor table and placed in a processor register. This must occur each time that SPICE accesses data in a different 64K region. For small circuits, this is not a problem, however, circuits on the order of 1000 nodes will span 16 or more segments with nearly every memory access requiring access to the descriptor table.